ECON 5350 Solutions to Problem Set #2

1. Greene 5th edition, Exercise 13.1.

   Answer:

   (a) See the attached gauss code.

   (b) See the attached gauss code. The critical value for the F test with a 5% significance level, 2 numerator degrees of freedom and 26 denominator degrees of freedom is 3.38. Therefore, we reject the null that the constant is the same for all three firms.

   (c) See the attached gauss code. The critical value for the LM test with a 5% significance level and 1 degree of freedom is 3.84. Therefore, we reject the null in favor of the RE model.

   (d) See the attached gauss code. The critical value for the Hausman test with a 5% significance level and 1 degree of freedom is 3.84. Therefore, we fail to reject the null that the explanatory variables are correlated with the errors, which lends credibility to the RE model.

2. Two-way fixed-effects model.

   Answer: See the attached gauss code. The critical value for the F test with a 5% significance level, 9 numerator degrees of freedom and 17 denominator degrees of freedom is 2.55. Therefore, we fail to reject that the period-specific effects are jointly zero.

3. Greene 5th edition, Exercise 13.8.

   Answer: The variance-covariance matrix is

$$\begin{bmatrix} \sigma^2+\sigma_\mu^2+\sigma_\nu^2 & \sigma_\mu^2 & \sigma_\nu^2 & 0 \\ \sigma_\mu^2 & \sigma^2+\sigma_\mu^2+\sigma_\nu^2 & 0 & \sigma_\nu^2 \\ \sigma_\nu^2 & 0 & \sigma^2+\sigma_\mu^2+\sigma_\nu^2 & \sigma_\mu^2 \\ 0 & \sigma_\nu^2 & \sigma_\mu^2 & \sigma^2+\sigma_\mu^2+\sigma_\nu^2 \end{bmatrix}.$$

4. Greene 5th edition, Exercise 13.12. Part (a).

   Answer: See the attached gauss code.

```
@ ************************ @
@ ECON 5350 PROBLEM SET #2 @
@ ************************ @

@ ************* @
@ EXERCISE 13.1 @
@ ************* @

@ Load Data @
load path = c:\gauss35\classes\econ5350\data\;
load data[31,4] = exer13.1;
y = data[2:31,3];
x = data[2:31,4];
trend = data[2:31,2];
T = 10;
n = 3;
k = 1;
constant = ones(n*T,1);

@ Rearrange Data @
y1 = y[1 4 7 10 13 16 19 22 25 28,1];
y2 = y[2 5 8 11 14 17 20 23 26 29,1];
y3 = y[3 6 9 12 15 18 21 24 27 30,1];
y = y1|y2|y3;

x1 = x[1 4 7 10 13 16 19 22 25 28,1];
x2 = x[2 5 8 11 14 17 20 23 26 29,1];
x3 = x[3 6 9 12 15 18 21 24 27 30,1];

t1 = trend[1 4 7 10 13 16 19 22 25 28,1];
t2 = trend[2 5 8 11 14 17 20 23 26 29,1];
t3 = trend[3 6 9 12 15 18 21 24 27 30,1];

x = x1|x2|x3;
trend = t1|t2|t3;
xmat = constant~x;
k = cols(xmat);

@ **************************** @
@ Part (a) -- Pooled Regression @
@ **************************** @

b = inv(xmat'*xmat)*(xmat'*y);
residp = y - xmat*b;
eep = residp'*residp;
varb = (eep/(n*T-k))*inv(xmat'*xmat);
tstat = b ./ diag(sqrt(varb));

print "Pooled OLS Estimates                tstats";
print b~tstat;
print;
```

Pooled OLS Estimates   tstats

```
   -0.74747578          -0.78191682
    1.0589589            18.053813
```

```
@ ********************************** @
@ Part (b) -- Fixed Effects Estimator @
@ ********************************** @

@ LSDV Approach @
D = zeros(n*T,n);
for ii(1,n,1);
for jj(1,T,1);
   D[(ii-1)*T+jj,ii] = 1;
endfor;
endfor;

xmat1 = xmat[.,2]~D;
k = k - 1;

@ FE Estimatior @
bfe = inv(xmat1'*xmat1)*(xmat1'*y);
residfe = y - xmat1*bfe;
eefe = residfe'*residfe;
varbfe = (eefe/(n*T-n-k))*inv(xmat1'*xmat1);
tfe = bfe ./ diag(sqrt(varbfe));

print "FE Estimates                    tstats";
print bfe~tfe;
print;
```

FE Estimates tstats

1.1021917	*21.731504*
-1.4684450	*-1.5356595*
-2.8361917	*-2.8222599*
0.12166184	*0.14194665*

```
@ Calculate F Statistic for Group Specific Effects @
F = ((eep - eefe)/(n-1))/(eefe/(n*T-n-k));
print "F statistic for significance of group specific effects = " F;
print;
```

F statistic for significance of group specific effects = 6.8109771

```
@ ********************************** @
@ Part (c) -- Random Effects Estimator @
@ ********************************** @

@ Calculate Estimate of Var(eps) @
Md = eye(n*T) - D*inv(D'*D)*D';
ystar = Md*y;
xstar = Md*xmat[.,2];
bfe1 = inv(xstar'*xstar)*(xstar'*ystar);
resid1 = ystar - xstar*bfe1;
s2e = (resid1'*resid1)/(n*T-n-k);

@ Calculate Estimate of Var(mu) @
Md1 = D*inv(D'*D)*D';
ystar = Md1*y;
```

```
xstar = Md1*xmat;
resid2 = ystar - xstar*b;
s2star = (resid2'*resid2)/(n-k-1);
s2u = s2star/T - s2e/T;

@ Feasible GLS @
theta = 1 - (sqrt(s2e)/(sqrt(T*s2u + s2e)));
Mdtheta = eye(n*T) - theta*Md1;
ystar = Mdtheta*y;
xstar = Mdtheta*xmat;
bre = inv(xstar'*xstar)*(xstar'*ystar);
residre = ystar - xstar*bre;
s2re = (residre'*residre)/(n*T-k-1);
varbre = s2re*inv(xstar'*xstar);
tre = bre ./ diag(sqrt(varbre));

print "RE Estimates                    tstats";
print bre~tre;
print;
```

RE Estimates   tstats

```
    -1.3414911          -0.99366658
     1.0986605           22.063602
```

```
@ LM Test @
lmnum = sumc(residp[1:T])^2 + sumc(residp[T+1:2*T])^2 + sumc(residp[2*T+1:3*T])^2;
LM = (n*T/(2*(T-1)))*((lmnum/eep)-1)^2;
print "Lagrange multiplier statistic = " LM;
print;
```

Lagrange multiplier statistic =   8.4720326

```
@ ************************ @
@ Part (d) -- Hausman Test @
@ ************************ @

W = ((bfe[1,1] - bre[2,1])^2)/(varbfe[1,1] - varbre[2,2]);
print "Hausman statistic = " W;
print;
```

Hausman statistic =   0.13433928

```
@ ************************************************* @
@ EXERCISE 13.1 CONT. -- TWO-WAY FIXED-EFFECTS MODEL @
@ ************************************************* @

@ Create New X Matrix @
DT = zeros(n*T,T-1);
for ii(1,T-1,1);
   DT[.,ii] = (trend .== ii);
endfor;
xmat2 = xmat1~DT;
k = cols(xmat2);

@ Two-Way FE Estimation @
```

```
btwfe = inv(xmat2'*xmat2)*(xmat2'*y);
residtwfe = y - xmat2*btwfe;
eetwfe = residtwfe'*residtwfe;
varbtwfe = (eetwfe/(n*T-k))*inv(xmat2'*xmat2);
ttwfe = btwfe ./ diag(sqrt(varbtwfe));

print "Two-Way Fixed-Effects Estimates             tstats";
print btwfe~ttwfe;
print;
```

Two-Way Fixed-Effects Estimates tstats

1.1109925	*16.332366*
-0.38116727	*-0.26444112*
-1.7591405	*-1.1771492*
1.2306512	*0.92015768*
-1.9939131	*-1.4156837*
-0.86736362	*-0.58777906*
-0.39614409	*-0.26249401*
-1.9182139	*-1.3158795*
-2.3073783	*-1.6260868*
-2.6903024	*-1.9134989*
-1.7823655	*-1.2321259*
0.21579518	*0.15281647*
-0.48795424	*-0.33834373*

```
@ F-Test for Period Effects @
R = zeros(T-1,4)~eye(T-1);
F = (R*btwfe)'*inv(R*varbtwfe*R')*(R*btwfe)/(T-1);
print "F Test for period-specific effects = " F;
print;
```

F Test for period-specific effects = 1.0806351

```
@ ******************** @
@ EXERCISE 13.12 PART A @
@ ******************** @

@ Load Data @
load data[31,4] = exer13.12;
y = data[2:31,3];
x = data[2:31,4];
T = 10;
n = 3;
constant = ones(n*T,1);

@ Rearrange Data @
y1 = y[1 4 7 10 13 16 19 22 25 28,1];
y2 = y[2 5 8 11 14 17 20 23 26 29,1];
y3 = y[3 6 9 12 15 18 21 24 27 30,1];
y = y1|y2|y3;

x1 = x[1 4 7 10 13 16 19 22 25 28,1];
x2 = x[2 5 8 11 14 17 20 23 26 29,1];
x3 = x[3 6 9 12 15 18 21 24 27 30,1];
```

```
x = x1|x2|x3;
xmat = constant~x;
k = cols(xmat);

@ Pooled Regression @
b = inv(xmat'*xmat)*(xmat'*y);
residp = y - xmat*b;
eep = residp'*residp;
varb = (eep/(n*T-k))*inv(xmat'*xmat);

@ Feasible GLS -- Groupwise Hd @
sig1 = ones(T,1)*(residp[1:T]'*residp[1:T]/T);
sig2 = ones(T,1)*(residp[T+1:2*T]'*residp[T+1:2*T]/T);
sig3 = ones(T,1)*(residp[2*T+1:3*T]'*residp[2*T+1:3*T]/T);
sig = sig1|sig2|sig3;
omega = zeros(n*T,n*T);
omega = diagrv(omega,sig);
bfgls = inv(xmat'*inv(omega)*xmat)*(xmat'*inv(omega)*y);
varbfgls = inv(xmat'*inv(omega)*xmat);
tfgls = bfgls ./ diag(sqrt(varbfgls));

print "Groupwise Heteroscedastic FGLS Estimates     tstats";
print bfgls~tfgls;
print;
```

Groupwise Heteroscedastic Feasible GLS Estimates tstats

7.1789347	*1.5033015*
1.1379160	*4.9913449*